
Namen anstelle von Zellbezügen 

Formeln, in denen Namen verwendet werden, sind verständlicher und übersichtlicher.

(Vorführen in fahrrad.xls)

a) Namen vergeben (EINFÜGEN / NAMEN / DEFINIEREN...)
· Markieren Sie die Zelle oder den Bereich, der benannt werden soll.

· Klicken Sie in das Namenfeld (links neben der Bearbeitungsleiste) und geben Sie den gewünschten Namen ein  (z.B. Dollarkurs oder Gesamtumsatz)

· <Return>. 


Hinweis: Im Dialogfeld Namen Definieren (Strg+F3) kann man ebenfalls neue Namen Hinzufügen. Außerdem kann man sich die definierten Namen mit zugehörigem Zellbezug ansehen.
Richtlinien für Namen:

Ein Name darf aus bis zu 255 Zeichen bestehen. Leerzeichen, Semikolon, Bindestrich und Doppelpunkt sind in Namen nicht erlaubt, als Trennzeichen zwischen Wörtern empfiehlt sich ein Punkt oder ein Unterstrich.

Das erste Zeichen eines Namens muss ein Buchstabe oder ein Unterstrich sein.

Automatische Namenvergabe (EINFÜGEN / NAMEN / ERSTELLEN...)

Excel bietet die Möglichkeit Spalten- bzw. Zeilenüberschriften im Tabellenblatt auch automatisch als Namen für die entsprechenden markierten Zellen übernehmen zu lassen.

· Markieren Sie die zu benennenden Zellen inklusive der Zeilen bzw. Spaltenbeschriftungen 

· dann EINFÜGEN / NAMEN / ERSTELLEN...(STRG+Shift+F3)

Im Namenfeld erscheinen die definierten Namen alphabetisch geordnet und können angesprungen werden, d.h. klickt man einen Namen an, so wird der zugehörige Bereich im Tabellenblatt markiert.

Beispiel: gesamte Tabelle markieren und Namen aus obersten Zeile und 1. Spalte übernehmen
b) Namen in Formeln einsetzen (EINFÜGEN / NAMEN / EINFÜGEN...,( <F3>))
Beginnen Sie mit der Formeleingabe bis zu der Stelle, an der ein Name einzusetzen ist. Öffnen Sie die Namensliste über den Befehl EINFÜGEN / NAMEN / EINFÜGEN... oder über die Funktionstaste <F3> und wählen Sie den gewünschten Namen aus. (Die Namen können auch über Tastatur eingegeben werden)

Beispiel:  Unter Umsatz eintragen:
= Verkaufszahlen * Stückpreis

Benutzen Sie in einer Formel einen Bereichsnamen und es wird ein einzelner Wert erwartet, so sucht Excel den Wert innerhalb der aktuellen Zeile bzw. Spalte aus diesem Bereich
Schnittmenge mehrerer Bereiche 

Will man explizit eine einzelne Zelle angeben, die im Schnittpunkt zweier benannter Bereiche liegt, so kann man das mit dem Schnittmengenoperator, dem Leerzeichen, zwischen den Bereichsnamen

Beispiel: =Gesamt Umsatz liefert die Schnittmenge von Gesamt und Umsatz, also den Gesamtumsatz.

c) Namen definieren, bearbeiten oder löschen

EINFÜGEN / NAMEN / DEFINIEREN...(<Strg>+<F3>)

Den gewünschten Namen markieren, dann den Geltungsbereich ändern oder den Namen löschen.

Namen verwenden standardmäßig absolute Zellbezüge.

Hinweis: Man kann hier auch Namen für eine konstante Zahl oder eine Formel definieren oder Namen, die sich auf Zellen in anderen Arbeitsmappen beziehen.

( Übung 1
d) Nachträglich Namen in Formeln übernehmen (EINFÜGEN / NAMEN / ÜBERNEHMEN...)
Sollten Sie vor der Namendefinition bereits Formeln erstellt haben, können Sie nachträglich die Bezüge in den Formeln automatisch durch Namen ersetzen lassen;  

· EINFÜGEN / NAMEN / ÜBERNEHMEN
· Namen auswählen, die in den vorhandenen Formeln eingesetzt werden sollen (z.B.  Produktionszahlen und Verkaufszahlen)
e) Verknüpfungen mit anderen Mappen über Namen
Sie können auch eine Verknüpfung mit einem definierten Namen in einer anderen Arbeitsmappe erstellen oder einen Namen definieren, der sich auf Zellen in einer anderen Arbeitsmappe bezieht. 
Vorführen: Beide Mappen fahrrad und dollarkurs öffnen.
· Namen aus einer anderen Mappe verwenden
In der Datei fahrrad Formel eingeben zur Berechnung des Stückpreis in $, dabei über FENSTER zu dollarkurs wechseln, F3 und den Namen Kurs_Feb04 auswählen. Also: =E5*dollarkurs.xls!Kurs_Feb04
· Namen definieren, der sich auf eine Zelle in einer anderen Mappe bezieht
In der Datei fahrrad Strg+F3 zur Definition eines neuen Namen, neuen Namen Dollarkurs04 eintippen, ins Eingabefeld BEZIEHT SICH AUF: klicken, über FENSTER zur anderen Mappe wechseln, Zelle anklicken und <Return>
1 Bedingte Formatierung
Die Formatierung einzelner Zellen lässt sich an Bedingungen knüpfen. Ist eine Bedingung erfüllt, wird das festgelegte Format automatisch der Zelle zugewiesen. Ist die Bedingung nicht  erfüllt, wird die ursprüng-liche Formatierung wiederhergestellt.

Formatierung abhängig vom ZELLWERT:

Beispiel 1: Wenn der Gesamtumsatz ZWISCHEN 500.000 UND 1.000.000  dann blau

· Vorführen: Gesamtumsatz markieren,  FORMAT / BEDINGTE FORMATIERUNG / 

ZELLWERT IST   zwischen 500.000 und 1.000.000
FORMAT ...
FARBE: Blau

       (zur Kontrolle mittendrin eine neue Zeile mit hohem Wert eingeben)

Beispiel 2: Lagerbestände < 15 sollen rot dargestellt werden
·   Gesamte Spalte markieren, FORMAT / BEDINGTE FORMATIERUNG / 

ZELLWERT IST  kleiner als  15


FORMAT ...   FARBE: Rot

Formatierung abhängig von FORMEL bzw. anderen Zellinhalten:

Beispiel 3: Siehe Datei provis.xls, Blatt Sprint
Die Namen der Schüler, deren Zeit <12, sollen rot dargestellt werden. Hierbei ist die Farbe vom Zellinhalt einer anderen Zell abhängig.


Zelle C3 auswählen, FORMAT / BEDINGTE FORMATIERUNG



FORMEL IST  =B3<12


FORMAT ...   FARBE: Rot

Anschließend das Format mittels Kopierpinsel nach unten kopieren.

Anstelle von ZELLWERT IST müssen Sie FORMEL IST verwenden, wenn die Formatierung vom Zellinhalt einer anderen Zelle abhängig ist (Beispiel 3) oder in der Bedingung etwas berechnet werden soll, beispielsweise wenn der größte Wert markiert werden soll.

FORMEL IST
=B5=MAX($B$5:$B$11)
Beispiel 4:

Die drei schnellsten Schülerzeiten und Namen sollen farblich markiert werden, der schnellste rot, der zweitschnellste lila und der drittschnellste blau. 

 z.B.

FORMEL IST  =B3=KKLEINSTE(Sprintzeit;2)
Hinweise: 

KKLEINSTE(Bereich; Rangnummer k) liefert die k-kleinste Zahl

Man kann bei der bedingten Formatierung maximal drei Bedingungen angeben!

( Übung 2.1 und 2.2 
Beispiel 5: In einem Kalender die Samstage und Sonntage in roter Schrift
· Spaltenüberschriften, in A1 Text Januar zentriert über 2 Spalten, mit Ausfüllkästchen nach rechts die restl. Monatsnamen erzeugen 

· dann in A2 das Datum 1.1.07 eintragen und nach unten bis zum 31.1.07 ausfüllen lassen

· dann in A2, FORMAT / BEDINGTE FORMATIERUNG / 

FORMEL IST   =WOCHENTAG(A2;2)>5   
FORMAT ...
FARBE: Rot
Format nach unten kopieren
· In B2 eintragen =A1, dann benutzerdefiniertes Format ttt, dann runterkopieren

Spalte markieren und wieder bedingte Formatierung wie oben

Hinweis: WOCHENTAG(Datum,Typ) liefert den Tag als Zahl 1 bis 7. Gibt man als Typ 2 an, so beginnt die Zählung mit dem Montag, so dass bei Samstagen eine 6 und bei Sonntagen eine 7 geliefert wird.

Formate entfernen   

Um alle Zellformate der markierten Zellen zu entfernen, zeigen Sie im Menü Bearbeiten auf Löschen, und klicken Sie dann auf Formate.

Suchen nach Zellen mit bedingten Formaten

Um nach Zellen mit bestimmten bedingten Formaten zu suchen, klicken Sie auf eine Zelle mit den zu suchenden Formaten, ansonsten klicken Sie auf eine beliebige Zelle.

Dann Menü Bearbeiten / Gehe zu / Inhalte / Bedingte Formate. 

Unter Gültigkeitsprüfung  Gleiche auswählen, um Zellen mit identischen bedingten Formaten zu finden, für alle bedingten Formaten  Alles.
Logische Funktionen

1.1 Die WENN-Funktion   Wenn(Bedingung;Dann_Wert;Sonst_Wert)
Man kann eine Berechnung von einer Bedingung abhängig machen, z.B. die Provisionen für die Mitarbeiter sind abhängig vom Umsatz.

Die Formel für die Provision (in provis.xls) zeigen: =WENN(Umsatz<200000;Umsatz*2%;Umsatz*4%)
WENN Umsatz kleiner 200.000  DANN 2% des Umsatzes SONST 4% des Umsatzes

Danach im Assistenten erst zeigen (Hinter Prüfung erscheint  FALSCH oder WAHR.) dann selber eingeben.
Zur Formulierung der Bedingung werden zwei Werte miteinander verglichen mittels Vergleichsoperatoren: = (gleich), <> (ungleich), < (kleiner), > (größer), <= (kleiner oder gleich), >= (größer oder gleich)

1.2 Verschachtelte Funktionen 

Als Argument einer Funktion können Sie auch das Ergebnis einer anderen Funktion verwenden.
Beispiel:  Mittelwertfkt. innerhalb der Wenn-Fkt, vorführen in pruefungen.xls
Wenn die Punktzahl einer Testperson besser ist als der Mittelwert, so soll das Wort „überdurchschnittlich“  erscheinen, andernfalls das Wort „unterdurchschnittlich“.

Sie können den Funktions-Assistenten verwenden, wenn Sie eine Funktion als Argument einer anderen Funktion verwenden wollen.
=>Vorführen 3 Mal, erst mit Assistenten und Zelladressen, dann mit Assistenten unter Verwendung eines
     Namen für die Punktzahl, dann ohne Assistent mit Quickinfo und Excel- Hilfe.

Schritt1: Klicken Sie im Assistenten in das Argumenten-Feld an die Stelle   , an die der Funktionsaufruf eingefügt werden soll 

Schritt 2: Klicken Sie auf den Pfeil   für die Funktionenliste und wählen die gewünschte Funktion aus.

Der Assistent öffnet nun das Dialogfenster für die verschachtelte Funktion, für die nun die Argumente eingegeben werden müssen.  In der Bearbeitungsleiste jedoch sieht man die komplette Formel, mit den verschachtelten Funktionsaufruf.        Achtung: absolute Adressen für Mittelwertfkt. oder Bereichsnamen

Achtung: Jetzt nicht auf [OK] oder Abbrechen klicken !!!

Schritt3: Um wieder in das Dialogfenster der übergeordneten Funktion zurückzukehren, klicken Sie in der Bearbeitungszeile innerhalb der Formel auf den Namen dieser Funktion (hier: auf WENN).
 (Wenn Sie auf OK drücken, wird die komplette Formeleingabe mittels Funktionsassistent beendet.)

Geben Sie nun weitere Argumente, wenn nötig, für die übergeordnete Funktion ein und beenden Sie die komplette Formeleingabe mit der Schaltfläche OK.

Formeln nachträglich im Assistenten bearbeiten

Wollen Sie bereits existierende Formeln korrigieren und im Assistenten bearbeiten, so aktivieren Sie zunächst die Zelle mit der Formel und klicken dann links neben der Bearbeitungszeile auf das Symbol [image: image5.bmp] (bei älteren Versionen auf das Gleichheitszeichen [image: image6.png]


).  
Um nun noch zwischen verschachtelten Funktionen im Funktions-Assistenten zu wechseln, klicken Sie in der Bearbeitungszeile auf den Namen der Funktion.

Formeln ohne Assistenten eingeben
Nachdem Sie in einer Formel den Funktionsnamen und eine öffnende Klammer eingegeben haben, wird ein Funktions-Quickinfo mit den notwendigen Argumenten eingeblendet (Excel 2002). Klicken Sie in diesem Quickinfo auf den Funktionsnamen, um das Hilfethema zu der Funktion anzuzeigen.
Beim Vorführen die Anführungszeichen bei den Textargumenten “überdurchschnittlich“ und “unterdurchschnittlich“ nicht vergessen.
Bei älteren Excel-Versionen können Sie sich die Argumente eintragen lassen, wenn Sie nach der Eingabe von =Funktionsname(   STRG+Shift+a  drücken. Drücken Sie nach Eingabe der öffnenden Klammer die Tastenkombination STRG+a, so startet der Funktionsassistent mit der eingegebenen Formel.
( Übung 3  (In 3.4 verschachtelte Wenn-Funktionen)
1.3 Bedingungen verknüpfen mit den Funktionen UND, ODER und NICHT 

Mehrere Bedingungen lassen sich durch die logischen Funktionen UND(), ODER() und NICHT() verknüpfen. Beispiel: Wer einen Umsatz > 300.000 hat ODER vor 1990 eingestellt wurde, bekommt als Geburtstagsgeschenk einen Walkman, ansonsten gibt’s nur einen Kugelschreiber. WENN(ODER(Umsatz>300000;Einstjahr<1990);"Walkman";"Kugelschreiber")   (in provis.xls)

1.4 Verschachtelte Wenn-Funktion

In Übung 2.4 sollten bereits zwei Wenn-Funktionen miteinander verschachtelt werden. Hier noch ein weiteres Beispiel:

=WENN(Umsatz<50000;0;WENN(Umsatz<200000;Umsatz*2%;Umsatz*4%))

WENN Umsatz kleiner 50.000 DANN keine Provision

 SONST    (WENN Umsatz kleiner 200.000 DANN 2% des Umsatzes SONST 4% des Umsatzes)

Sie können bis zu sieben WENN()-Funktionen ineinander verschachteln und somit maximal acht Fälle unterscheiden. 

( Zusatzaufgabe
Nachschlagen mit SVERWEIS und WVERWEIS

Anstelle einer mehrfach verschachtelten WENN-Funktion, benutzen wir nun die SVERWEIS-Funktion. Diese ermöglicht es, in einem festzulegenden Bereich nach einem bestimmten Wert zu suchen und dann einen Text oder eine Zahl, die neben dem gefundenen Wert abgelegt ist, zurückzugeben.

In provis.xls (Provision mit SVERWEIS) wird eine sortierte Provisionstabelle als Suchbereich definiert. 
Allgemein:


=SVERWEIS(Suchwert;Suchbereich;Ergebnisspalte) 

Im Beispiel:

=SVERWEIS(Umsatz; Provisions_Tabelle; 2)

Achtung: Den Suchbereich mit absoluten Adressen angeben oder den Namen.

Wie funktioniert die Verweisfunktion SVERWEIS()?

-
Excel durchsucht die 1. Spalte des Suchbereichs von oben nach unten nach dem als Suchkriterium angegebenen Wert 

-
Findet Excel den Wert, so wird der rechts daneben in der Ergebnisspalte stehende Wert als Ergebnis zurückgeliefert.

-
Kann Excel keinen Wert  finden, der exakt mit dem Suchkriterium übereinstimmt, so wird der nächstkleinere gezeigt (bei sortierten Bereichen).

-
Ist der Suchwert kleiner als der erste Eintrag des Suchbereichs, erscheint der Fehlerwert #NV. 

Beispiele: siehe Porto und in Sprint die 3 ersten Plätze(in provis.xls), siehe auch Excel-Hilfe.
Anmerkungen zu den Verweis-Funktionen:
· SVERWEIS(Suchkrit.;Matrix; Ergebnisspaltenindex;Verweistyp) sucht senkrecht in der 1. Spalte der Matrix und liefert aus der gleichen Zeile einen Wert aus der Ergebnisspalte

· WVERWEIS(Suchkrit.;Matrix; Ergebnisspaltenindex;Verweistyp) sucht waagerecht in der 1. Zeile der Matrix und liefert aus der gleichen Spalte einen Wert aus der Ergebniszeile.
· Ist der Bereich in dem gesucht wird, nicht sortiert, sollte als letztes Argument in diesen Funktionen für den Suchtyp eine 0 angegeben werden. Dann werden bei der Suche nur genaue Entsprechungen zugelassen.
( Übung 4, Übung 5 und Zusatzaufgaben
2 Suchen mit VERGLEICH
Die VERGLEICH-Funkion liefert im Gegensatz zu den VERWEIS-Funktionen keinen Wert aus einer Zelle sondern die Position des gefundenen Wertes innerhalb des Suchbereichs. VERGLEICH(Suchkrit.; Suchbereich;Vergleichstyp)     Vergleichstyp ist die Zahl -1, 0 oder 1.

Beispiel 1: 

	Land


	
	Spanien
	Belgien
	Finnland
	Frankreich
	Holland

            Währung 

  Datum
	Span.
Peseta
	Belg.
Franc
	Finnmark
	Franz.
Franc
	Holl.
Gulden

	21.9.1998
	81,5549
	18,3235
	3,02110
	3,06831
	1,07603

	22.9.1998
	82,3787
	18,5086
	3,03019
	3,06524
	1,08253

	23.9.1998
	82,6266
	18,5643
	3,01212
	3,04696
	1,07607

	24.9.1998
	83,1254
	18,7518
	3,00311
	3,05613
	1,08694

	25.9.1998
	83,9651
	18,6400
	2,98520
	3,05308
	1,08369

	28.9.1998
	83,7140
	18,6961
	2,98819
	3,05003
	1,08695

	29.9.1998
	83,6304
	18,7148
	2,97037
	3,01983
	1,08586


=VERGEICH("Belgien";A1:A6;0)  liefert als Ergebnis 3, da dritte Zelle im Suchbereich A1:A6

In der Devisentabelle (etwas veraltet aber sehr lehrreich) soll der Wechselkurs nachgesehen werden, der abhängig vom Datum und vom Land ist. Mit der sverweis-Funktion sucht man dann das Datum und holt sich den Kurs aus der Spalte, die dem gewünschten Land entspricht. Die richtige Spalte erhält man mittels der VERGLEICH-Funktion, die in dem Bereich mit dem Namen Spaltenüber-schriften nach dem Land sucht und dessen Position liefert.

=SVERWEIS(A3;Kurse;VERGLEICH(D3;Spaltenüberschriften;0)))

Beispiel 2: (siehe Blatt „INDEX und VERGLEICH“)
Kann man die SVERWEIS-Funktion nicht verwenden, weil die Suchspalte  rechts von der Ergebnisspalte liegt, so könnte man die Anordnung in der Tabelle verändern oder mittels VERGLEICH in der Suchspalte die Position ermitteln und dann in der Ergebnisspalte an der gewünschten Position mit INDEX nachsehen.

Auf unserem Tabellenblatt Sprint wurden die Namen der drei besten Läufer ermittelt mit

=SVERWEIS(F3;Sprintliste;2;0)
	E
	F
	G

	
	Laufzeit
	Sieger

	1. Platz
	11,48
	Campo  Pia

	2. Platz
	11,50
	Bauhans Krimhild

	3. Platz
	11,59
	Pastelink  Monika


Angenommen in unserer Sprintliste stünde die Spalte Zeit rechts von Spalte Name. Dann müsste man entweder die beiden Spalten vertauschen, um die SVERWIS-Funktion verwenden zu können, oder:
=INDEX(Name;VERGLEICH(F3;Zeit;0))
Hinweis: INDEX(Bereich;3) liefert den dritten Wert des Bereichs.
Die Funktionen SUMMEWENN und ZÄHLENWENN

Bei den VERWEIS- und VERGLEICH-Funktionen wurde mit dem Suchkriterium ein Wert in einem Bereich gesucht. Nun sollen alle Werte, die dem Suchkriterium entsprechen, gezählt werden.

2.1 ZÄHLENWENN und SUMMEWENN 
Dateien: personal.xls/Demo und fahrrad.xls
1) Zählen wie viele Datensätze ein best. Kriterium erfüllen

ZÄHLENWENN(Bereich;Suchkriterien)  zählt wie viele Zellen bzw. Zeilen Bereich enthält, die den Suchkriterien entsprechen.



Demo1:

Anzahl der Mitarbeiter in der Marketing-Abteilung
=ZÄHLENWENN(F2:F26;"Marketing")

(Den Bereich F2:F26 entweder markieren oder Bereichsnamen verwenden z.B. Abteilung)
Personen älter als 50:
=ZÄHLENWENN(Alter;“>50“)
Achtung: Die Suchkriterien werden in Anführungszeichen eingeschlossen 

Der Bereich wird als Zellbereich z.B. B4:B23 angegeben bzw. sein Name.

2) Addieren (summieren) von Werten der Datensätze, die ein best. Kriterium erfüllen

SUMMEWENN(Bereich;Suchkriterien;Summe_Bereich)  addiert nur die Zahlenwerte aus Summe_Bereich, die den Suchkriterien in Bereich entsprechen.

Gehaltssumme in der Abteilung Marketing:
=SUMMEWENN(F2:F26;"Marketing“;D2:D26)
(Der erste Bereich ist der Suchbereich, der Summe_Bereich enthält die zu addierenden Zahlen
Zum Demonstrieren am besten Tabelle nach Abteilung sortieren, die Abteilung Marketing grau markieren und die zugehörigen in die Summe eingehenden Gehälter gelb markieren).


( Übung 6
3) Vorführen, Beispiel Suchbereich und Summierungsbereich identisch:

Demo 1:

Summe der pos Zahlen:   =SUMMEWENN(Bereich;“>0“;Bereich) bzw. =SUMMEWENN(Bereich;“>0“)
Summe der neg. Zahlen:  =SUMMEWENN(Bereich;“<0“) bzw. =SUMMEWENN(Überschrift;"<0")
Hinweis: Anstelle z.B. des Bereichs B36:B41 kann man auch (wenn explizit kein Name definiert wurde), die Überschrift z.B. Zahlenfolge verwenden (vorausgesetzt, sie kommt in der Mappe nur einmal vor).
4) Suchkriterien für Zählenwenn und Summewenn

i.a. einfache Ausdrücke in Hochkommata  ”>1000”  oder ”Meier”  

(erlaubte Operatoren in den Ausdrücken:   =     >     <      >=      <=     <>)

Sie können die Suchkriterien flexibler gestalten durch Verwendung 

· des Platzhalterzeichens *

· von Zelladressen

· von berechneten Werten

Beispiele:

· =ZÄHLRNWENN(Beruf;“A*“) gezählt werden die Angestellten und Arbeiter
=SUMMEWENN(Artikel;"*Damen*";Umsatz)  in fahrrad.xls
                                              summiert werden die Umsätze der Damenfahrräder

· Die Abteilung, deren Mitarbeiter gezählt werden soll, soll variabel sein (Demo2). 
In Zelle L6 steht das Suchkriterium

In L7: =ZÄHLENWENN(F7:F31;L6)
 

· Zählen wie viele Mitarbeiter sind überdurchschnittlich alt:

      In  L9 wird das Durchschnittsalter berechnet.
 
In L10: =ZÄHLENWENN(H7:H31;">"&L9)
        =ZÄHLENWENN(H7:H31;">"&MITTELWERT(H7:H31))
( Zusatzaufgabe
2.2 Duplikate anzeigen mit  ZÄHLENWENN 

· Erstellen Sie am Ende der Liste eine Spalte namens Duplikate.

· Benutzen Sie nun die Funktion ZÄHLENWENN, um zu zählen, wie oft z.B. eine Mitgliedernummer in der Spalte A vorkommt. =ZÄHLENWENN($A$26:$A$404;A26) 

· Mit dem Autofilter lassen sich nun die Datensätze bequem herausfiltern, die in der Spalte Duplikate eine Zahl >1 enthalten DATEN / FILTER / AUTOFILTER 
In der Liste von Duplikate   (BENUTZERDEFINIERT ..) wählen und  ist grösser als 1. 

· Anstelle des Autofilters können Sie auch die Duplikate mittels bedingter Formatierung farbig markieren, Menü FORMAT / BEDINGTE FORMATIERUNG - Formel ist =$D26>1 

2.3 Duplikate verhindern mit DATEN / GÜLTIGKEIT und  ZÄHLENWENN 
Mit Gültigkeitsprüfungen kann man festlegen welche Einträge in bestimmten Zellen erlaubt (gültig) sind und welche nicht. Angenommen in der Spalte A sollen keine Duplikate (doppelte Einträge) vorkommen. Dann gehen Sie folgendermaßen vor:

· Markieren Sie die Spalte A  (oder nur A1, dann muss der Inhalt hinterher kopiert und mittels Inhalte einfügen / Gültigkeit in die anderen Zellen eingefügt werden)
· Unter DATEN / GÜLTIGKEIT im Register EINSTELLUNGEN wählen Sie unter
ZULASSEN:  Benutzerdefiniert         und unter 
FORMEL:
= ZÄHLENWENN(A:A;A1)=1
Die Formel liefert einen Wahrheitswert (WAHR für gültige und FALSCH für ungültige Einträge). 

· Dann im Register FEHLERMELDUNG  unter
TYP: Stopp        und  unter
FEHLERMELDUNG: Diesen Wert gibt es bereits einmal!!!
Versucht der Anwender nun in Spalte A ein Duplikat einzutragen, erscheint Karl Klammer mit der entsprechenden Fehlermeldung.
Hinweis: Mittels Gültigkeitsprüfung wird nur das Eintragen neuer Duplikate verhindert, nicht jedoch bereits vorhandene oder das Kopieren von doppelten Einträgen.


Wesentlich besser ist eine selbst geschriebene VBA-Ereignisprozedur,  
Sub Worksheet_Change(ByVal Target As Range), die Duplikate sucht und verhindert.
2.4 KKLEINSTE und ZÄHLENWENN 
Mit  KKLEINSTE(Bereich;Rangnummer) kann man z.B. den drittkleinsten Wert in einem Bereich ermitteln. Will man aus einer Liste das billigste Buch, das nicht umsonst ist, herausfischen, dann lautet die Formel =KKLEINSTE(A2:A9;2). Dabei steht die 2 für den Rang, also das zweitbilligste Buch. Mit einer 1 hätte man das Buch bekommen, das gar nichts gekostet hat. 
Was aber passiert, wenn mehrere Werte gleich sind? 

Man merkt sehr schnell, dass Excel jeden Wert einzeln zählt, so dass unsere obige Formel als zweitbilligstes Buch 0 € ergeben würde. Im Prinzip ist der Preis, den wir finden wollen der fünftkleinste Wert, da  vier Nullen vorhanden sind.

Und nun der Trick : Wenn man ZÄHLENWENN() mit KKLEINSTE() kombiniert, kommt man zu dem gewünschten Ergebnis: =KKLEINSTE(A2:A9;ZÄHLENWENN(A2:A9;0)+1)
Datenbank-Funktionen DBSUMME und DBANZAHL

Bei einfachen Suchkriterien (Bedingungen) reichen SUMMEWENN() und ZÄHLENWENN(), bei komplexen Kriterien nimmt man DBSUMME() und DBANZAHL(). Die Kriterien werden dabei nicht direkt in der Formel eingetragen, sondern in einen Zellbereich und in der Formel gibt man diesen Zellbereich für die Suchkriterien ein.

Demo 3  Wieviele Mitarbeiter sind im Vertrieb und wurden nach 1990 eingestellt ?

	Suchkriterien:
	Abteilung
	Einst_Jahr

	
	
	Vertrieb
	>1990


 In den Zellen D67:E68

=DBANZAHL(Datenbank;[Datenbankfeld];Zellbereich mit Suchkriterien)
=DBANZAHL(A1:I26;;D67:E68)

In Kapitel 7.6  (Demo 4) werden wir noch eine weitere Alternative für SUMMEWENN() und ZÄHLENWENN() kennen lernen, bei der man auch komplexere Kriterien verwenden kann.
Nämlich Matrixformeln, in denen SUMME() und WENN() verschachtelt werden.
Hinweis: 
Mit ZählenWenn() und SummeWenn() wird nur ein Ergebnis berechnet und zwar für die Datensätze, die bestimmte Kriterien erfüllen (also für eine best. Gruppe).

Bei den automatischen Teilergebnissen werden für jede Gruppe (Datensätze mit gleichem Eintrag in einem Feld) Ergebnisse automatisch berechnet, also für alle Datensätze.

Automatische Teilsummen (Teilergebnisse) und  Pivot-Tabellen wurden im Kurs Excel Fort gezeigt. Pivot-Tabellen (Kreuztabellen) geben ebenfalls Teil- und Gesamtergebnisse aus, verfügen aber über wesentlich mehr und flexiblere Möglichkeiten. Die einzelnen Datensätze werden allerdings nicht angezeigt.
3 Matrixformeln und  -funktionen

3.1 Matrixformeln 

Matrixformeln haben als Argumente Zellbereiche (Matrizen), das Ergebnis kann ein Bereich oder ein einzelner Wert sein.

Mit Matrixformeln können Sie mehrere Berechnungen gleichzeitig durchführen. Sie erleichtern das Erstellen von gleichen Formeln für benachbarte Zellen (Beispiele in Matrixformeln.xls.)
Zum Erstellen von Matrix-Formeln drückt man nach Eingabe der Formel in der Bearbeitungs​zeile Strg+Shift+Enter. Man erkennt Matrixformeln an den geschweiften Klammern.

{=Funktion(Argumente als Matrixbereiche)} oder {=Berechnungen mit Bereichen}  

Beispiele: Tagesumsatz der Fluggesellschaft, Durchschnittsnote der Klassenarbeit

Beispiele:





Berechnung:  

In drei Zellen:

{= A1:C1 * A10:C10}  


                                                                     

In einer Zelle:

{=SUMME(B10:E10*B11:E11)}         
 (B10*B11)+(C10*C11)+(D10*D11)+(E10*E11)
Achtung: Liefert eine Matrixformel mehrere Werte, muss vor Eingabe der Formel die entsprechende Anzahl an Zellen markiert werden.

Hat man die Formeleingabe aus Versehen mit Enter beendet, kann man auch nachträglich in der Bearbeitungszeile die Tastenkombination Strg+Shift+Enter eingeben.

3.2 Matrixkonstante  

Anstatt eines Zellbereichs kann man auch konstante Werte in einer Matrixformel eingeben. Dies ist sinnvoll, wenn mit Zahlen gerechnet werden soll, die sich nicht in der Tabelle befinden. 

Angenommen die Flugpreise befinden sich nicht wie in dem Beispiel oben in den Zellen B11:E11 , sondern sollen als Matrixkonstante eingegeben werden, dann müsste man anstelle {=SUMME(B10:E10*B11:E11)} schreiben  {=SUMME(B10:E10*{870.720.690.470})}
Anstelle der relativen Zellbezüge werden konstante Werte in geschweifte Klammern gesetzt und durch Semikolon bzw. Punkt getrennt. Werte, die eigentlich in verschiedenen Spalten stehen müssten, werden durch Punkte getrennt, und Werte, die eigentlich in verschiedenen Zeilen stehen müssten, werden durch Semikolon getrennt.

3.3 Matrix markieren und löschen

Um eine Matrixformel zu bearbeiten oder zu löschen, müssen sie vorher den gesamten Matrixbereich markieren mit Strg+Shift+/  oder

BEARBEITEN / GEHE ZU / [INHALTE] / AKTUELLES ARRAY

( Übung 7
3.4 Matrixfunktionen und Mathematische Matrizenrechnung
Man sieht, in Excel bedeutet Rechnen mit Matrizen, Rechnen mit Bereichen. 

Dies hat nichts mit der mathematischen Matrizenrechnung zu tun, dafür gibt es in Excel die Funktionen: MMULT, MINV  und MDET.  

MINV und MDET werden dazu verwendet lineare Gleichungssysteme mit mehreren Unbekannten zu lösen. (Beispiel in Matrixformeln.xls, Blatt: lin. Gleichungssystem)

Diese drei mathematischen Funktionen sind typische Matrixfunktionen ebenso MTRANS und die meisten Statistikfunktionen wie z.B. TREND und HÄUFIGKEIT.

Dagegen sind die meisten Funktionen unter der Kategorie Matrix (z.B. VERWEIS; SVERWEIS, WVERWEIS, VERGLEICH, INDEX) keine typischen Matrixfunktionen, da sie nicht mit der Tastenkombination Strg+Shift+Enter und somit ohne geschweiften Klammern eingegeben werden.

Hinweis: Matrixfunktion HÄUFIGKEIT und Statistische Analysefunktion HISTOGRAMM wurden bereits im Kurs Excel Fort gezeigt.
3.5 Matrixfunktion TREND
Trendanalysen werden gewöhnlich für Prognosen über zukünftige Entwicklungen aufgrund von Datenreihen der Vergangenheit erstellt. 

Vorführen in trend.xls (Besuchertrend, Umsatztrend, Leistungstrend, Trendlinien)

Beispiel:

Bekannt sind die Besucherzahlen des Freizeitparks für die Jahre 2002-2007.

Davon ausgehend, dass die Besucherzahlen in Zukunft ähnlich steigen, sollen nun die Besucherzahlen für die Jahre 2008-2012 prognostiziert werden.
Einfache Trendanalyse mit dem Ausfüllkästchen

Vorführen in trend.xls (Besuchertrend), mit dem Ausfüllkästchen Reihe erweitern.

Die bestehende Datenreihe wird erweitert indem ein berechneter Durchschnittswert jeweils dazu addiert wird (Linearer Trend). Man kann auch mit rechter Maustaste das Ausfüllkästchen „ziehen“ und im Kontextmenü  Exponentieller Trend wählen, dann wird mit einem best. Wert jeweils multipliziert.

Häufig will man aber nicht nur zukünftige Werte prognostizieren, sondern auch die Trendwerte für die vorliegenden Ergebnisse berechnen. Um z.B. zu sehen, ob die ausgewählte Trendfunktion sinnvolle Ergebnisse liefert, vergleicht man die berechneten Werte mit den gemessenen. 
Dies kann man schön in einem Diagramm veranschaulichen: Lineare Trendlinie im Diagramm mit Gleichung erstellen. 
Die Trendanalyse mittels Ausfüllkästchen ist nicht geeignet für komplexe oder nicht lineare Daten. 

Trendanalysen mit der Statistik-Funktion TREND  (Linearer Trend) 

Für einen linearen Trend gibt es die Funktion TREND(Y_Werte; X_Werte; Neue_x_Werte). Als Ergebnis liefert die Funktion die auf einer Geraden liegenden y-Werte, die zu den in Neue_X_Werte angegebenen x-Werten gehören.

Hinweis:

Die Funktion RGP liefert die Regressionsparameter für den linearen Trend. Dazu gehören die Steigung m und der y-Achsenabschnitt b für die Geradengleichung: y = m*x + b. Außerdem werden noch weitere Regressionskenngrößen wie Bestimmtheitsmaß r², Standardfehler, … geliefert.
Die Funktion VARIATION ist die exponentielle Entsprechung zur linearen Funktion TREND. Sie liefert Werte, die sich aus einem exponentiellen Trend ergeben und auf einer Exponential-kurve (z.B. y= b*m^x ) liegen.
Die Funktion RKP  liefert die zugehörigen Regressionskenngrößen u.a. ein WerteArray, das diese Kurve beschreibt.
Achtung: Trend ist eine Matrixfunktion, die mehrere Ergebnisse liefert. Man muss vorher die entsprechende Anzahl Zellen markieren. Die Anzahl muss der Anzahl der Werte in Neue_x_Werte  entsprechen.

1. Beispiel  (Besuchertrend mit X_Werte: 1998-2003 und Neue_X_Werte: 1998-2008)
=> Vorführen und TREND aufrufen
=> Dann Trendzahlen mit Überschrift ins Diagramm ziehen (den äußeren Rahmen der Markierung)
        Man sieht die Trendlinie, eine Gerade.

Fehlt die Matrix Neue_x_Werte (das 3. Argument), wird angenommen, dass sie mit der Matrix X_Werte  identisch ist.
In Beispiel 1 beziehen sich die Ergebniswerte nur auf regelmäßige Zeitintervalle, also keine echten x-Werte. Man spricht hier auch von Zeitreihen.

2.  Beispiel   GeschwindigkeitsTrend (in Tabelle Leistungstrend) mit echten x-Werten
TREND(Y_Werte; X_Werte; Neue_x_Werte)
{=TREND(Geschwindigkeit; Gewicht u. Alter; Gewicht u. Alter für die ein Ergebnis berechnet werden soll)}
Y_Werte: 
Als erstes Argument wird der Zellbereich der bereits bekannten Ergebniswerte angegeben.

X_Werte: 
Der  Zellbereich von dem die bekannten Ergebniswerte abhängig sind 
 

(z.B. die Geschwindigkeit der Läufer ist vom Gewicht und Alter abhängig.)



Neue_x_Werte: die  x-Werte, für die die Funktion TREND die zugehörigen y-Werte liefern soll.


(z.B. Gewicht und Alter der noch nicht gelaufenen Personen)


Sind die Y-Werte nur vom gemessenen Zeitpunkt abhängig, hat man keine echten X-Werte. 

Bei regelmäßigen Zeitintervallen spricht man dann von Zeitreihen und hat im Prinzip die 
X-Werte 1,2,3,4,5,6,...
In diesem Fall könnte man auch anstelle der Trend-Funktion das Ausfüllkästchen verwenden.

3.  Beispiel (Umsatztrend, ohne X-Werte bzw. 1, 2, 3,...)  

Vorführen: Umsatztrend für Juli – Sept. , dazu TREND aufrufen und 

a) für die X_Werte die Zahlen 1, 2, ...,6  und für Neue_x_Werte die Zahlen 7, 8, 9 markieren
b) keine x-Werte eingeben und für Neue_x_Werte wieder 7, 8 , 9 markieren.
c) keine x-Werte eingeben und für Neue_x_Werte die Matrixkonstante {7.8.9} eintippen
Hinweis: Siehe auch Datei Regression.xls
Beispiel: linearer Zusammenhang zwischen Zellgröße und Chloroplastenzahl beim Moos

Demo: Trendlinie, Matrixfunktion TREND und RGP und AddIn Analyse-Funktion REGRESSION und Funktion Trend_linear aus AddIn xlAutomationTrend.
Trendlinien mit Trendgleichung im Diagramm

Excel bietet die Möglichkeit, Trendlinien auf Regressionsbasis (Methode der kleinsten Quadrate) für fünf verschiedene Kurventypen und einen Trend als gleitende Mittelwerte zu berechnen und ins Diagramm zeichnen zu lassen. 
· Datenreihe im Diagramm markieren, 

· dann im Kontextmenü TRENDLINIE HINZUFÜGEN auswählen

· dann den Trendlinientyp auswählen 

· Unter OPTIONEN können Sie angeben, 
□  Gleichung im Diagramm darstellen
□  Bestimmtheitsmaß im Diagramm darstellen.

Beispiel: in Trendlinien Diagramm erstellen mit Arbeitslosenquote und Jahreszahlen
Datenreihe Arbeitslosenquote markieren, Kontextmenü TRENDLINIE HINZUFÜGEN...

TYP / Polynomisch  6. Ordnung und OPTIONEN / Formel im Diagramm darstellen


Evtl. Trendlinie formatieren und MUSTER / Farbe rot und Datenbeschriftung formatieren 


SCHRIFT / Farbe rot

Die Verlässlichkeit einer Trendlinie ist am größten, je näher ihr Bestimmtheitsmaß R2  bei 1 liegt.

Der Wert von R2  liegt zwischen 0 und 1

Hinweis

Das Add-In  xlAutomationTrend unter http://rtsoftwaredevelopment.de liefert für alle Trendlinien die zugehörigen mathematischen Tabellenfunktionen (Trend_linear, Trend_exponentiell, Trend_polynomial,...) zur Trendberechnung.
Ältere Variante: Trend2k
3.6 Verschachteln von SUMME und WENN

Sind die Suchkriterien für die Funktionen ZÄHLENWENN und SUMMEWENN (siehe Kapitel 4) zu komplex, so kann man SUMME und WENN in einer Matrixformel verschachtelten. 
Demo 4 in personal.xls
Summe der positiven Zahlen: 

{=SUMME(WENN(J73:J78>0;J73:J78;0))}
Anzahl der Mitarbeiter im Vertrieb: 
{=SUMME(WENN(F2:F26="Vertrieb";1;0))}
Anzahl der Mitarbeiter im Vertrieb und nach 1990 eingestellt:

{=SUMME(WENN(F2:F26="Vertrieb";WENN(I2:I26>1990;1;0);0))}
Gehaltssumme der Mitarbeiter im Vertrieb und nach 1990 eingestellt:

{=SUMME(WENN(F2:F26="Vertrieb";WENN(I2:I26>1990;D2:D26;0);0))}

Alternative: Der Teilsummen-Assistent:

Wem das zu schwierig ist, der verwendet den Teilsummen-Assistenten, der genau solche verschachtelte Matrixformeln erzeugt.

· Assistenten aufrufen: EXTRAS / ASSISTENT / TEILSUMMEN ...

· Schritt1: gesamten Bereich incl. Spaltenüberschriften angeben,

· Schritt2: Summierungsspalte auswählen und Bedingungen hinzufügen

Hinweis: 
Soll nicht die Summe, sondern die Anzahl berechnet werden, 
wählt man als Summierungsspalte eine Spalte mit Text.

· Schritt 3: Auswählen, ob nur die Formel oder auch die Bedingungswerte eingetragen werden sollen.

· Schritt4: Zelle für die Summierungsformel auswählen.

4 INDIREKT und Adressreihen

Mit der Funktion INDIREKT kann der Zellbezug einer Formel aus dem Inhalt einer oder mehrer Zellen erzeugt werden. Angenommen in B2  steht eine Zelladresse z.B. A14, dann liefert

=Indirekt(B2)   
den Inhalt der Zelle A14
Beispiele siehe indirekt.xls
Beispiel 1:  (Zelladresse in Zelle)
jeden dritten Wert auflisten

Man kann in Excel mit dem Ausfüllkästchen zwar Adressreihen z.B. A1, A4, A7, A10, ... erzeugen nicht aber in Formeln, also nicht =A1, =A4, =A7  bzw. =Funktion(A1), =Funktion(A4), = Funktion(A7). Beim Kopieren von Formeln mit dem Ausfüllkästchen werden die relativen Zellbezüge lediglich angepasst i.a. um 1 erhöht: =Formel(A1), =Formel(A2), =Formel(A3), ...
Will man nun jeden dritten Wert aus Spalte A auflisten, so erzeugt man in einer zusätzlichen Spalte mit dem Ausfüllkästchen die gewünschte Adressreihe (ohne =) und greift darauf mit der Funktion INDIREKT zu.
	A
	B
	C
	D

	Wert
	Jede 3. Zelle
	INDIREKT
	

	2
	A14
	2
	=Indirekt(B14)

	4
	A17
	8
	=Indirekt(B15)

	6
	A20
	14
	=Indirekt(B16)

	8
	A23
	20
	=Indirekt(B17)

	10
	A26
	26
	=Indirekt(B18)


Hinweis:

In dem Beispiel oben wurde die  Funktion INDIREKT entsprechend ihrem Namen für indirekte Adressierung verwendet- das Argument enthält dann keine Hochkommata.
Man kann die Funktion aber auch für einen ganz anderen Zweck verwenden. Möchten Sie, dass eine Formel immer auf dieselbe Zelle verweist, ungeachtet dessen, ob die Zeile über der Zelle gelöscht oder die Zelle verschoben wird, verwenden Sie INDIREKT und verwenden als Argument die Zelladresse eingeschlossen in Hochkommata
Den Inhalt der Zelle A14 erhält man mit:
a) = A14



(flexibel)
b) =Indirekt("A14") 


(starr)
c) =Indirekt("A" & 14) 

(starr)

Wird die Formel =A14 oder =Funktion(A14) nach unten kopiert, wird entsprechend die Zeilennummer erhöht. Wird die Zelle A14 verschoben, wird die Formel auch entsprechend angepasst. Nicht jedoch bei Funktion(Indirekt("A14"))
Beispiel 2 (ZeilenNr in Zelle)
Sie wollen von den täglich erfassten Messwerten, jeweils die durchschnittlichen Wochenwerte.

	A
	B
	C
	D

	Datum
	Messwert
	Startzeile
	Wochen Mittelwert

	1.1.2006
	0,00
	2
	1,23

	2.1.2006
	0,30
	9
	2,26

	3.1.2006
	0,30
	16
	0,80

	4.1.2006
	5,80
	23
	0,99

	5.1.2006
	0,00
	30
	0,96


                                         =MITTELWERT(INDIREKT("B" & C2):INDIREKT("B" & C2+6))
Beispiel 3 (Blattname in Zelle)
Für jeden Mitarbeiter gibt es ein separates Tabellenblatt, das mit der jeweiligen Personalnummer z.B. A-001 benannt wurde. Nun sollen auf einem neuen Übersichtsblatt für alle Mitarbeiter einzelne Werte z.B. Krankheits- und Urlaubstage aufgelistet werden.
Stehen nun auf dem Übersichtsblatt in Spalte A die Personalnummer also die Blattnamen, so kann man mit  =INDIREKT("'" &A2& "'!D2")  dann die zugehörigen Daten erhalten.
Verknüpfungen und Datenkonsolidierung

Im Grundlagenkurs haben wir bereits die Verknüpfung von Tabellen bzw. Arbeitsmappen kennen gelernt. Durch Eingabe eines Gleichheitszeichens und einer Zelladresse, wird ein Verweis auf den entsprechenden Zellinhalt erzeugt, auch wenn er sich in anderen Arbeitsmappen befindet.

Vorführen in Konsolidieren.xls: 

1. Manuelle Verknüpfungen

Vorführen in Gesamt 
für Haushaltsgeräte:  =Hamburg!B5+München!B5+Berlin!B5
   


für Verschiedenes:    =Hamburg!B7+München!B8+Berlin!B9
Werden die Quelldaten geändert, so werden die Daten bei Verknüpfungen natürlich automatisch aktualisiert – nicht so bei kopierten Daten.

Im Grundlagenkurs haben wir die 3D-Summen für gleichartig aufgebaute Tabellen kennen gelernt.

Nun haben wir aber unterschiedlich aufgebaute Tabellen.

2. Daten nach Rubrik konsolidieren

Konsolidierung bedeutet, dass Daten aus mehreren Tabellen zur Auswertung in einer Tabelle zusammengefasst und übersichtlich dargestellt werden.  Sind diese Tabellenblätter unterschiedlich aufgebaut, soll Excel die Zeilen- und Spaltenüberschriften als Grundlage nehmen und nur die Daten mit gleichen Überschriften (Rubriken) zusammenfassen. 

Zeigen: in Gesamt Ergebnis Konsolidierung nach Rubrik und dynamische Konsolidierung nach Rubrik .

a) statische Konsolidierung 

Die Ergebnisse in der neuen Zieltabelle sind konstante Zahlen, keine Formeln.
Linke obere Ecke in der neuen Tabelle auswählen, dann DATEN / KONSOLIDIEREN ...

     

( gewünschte Funktion auswählen

( Hinzufügen der Quellbereiche 
  (incl. der Zeilen und Spaltenüberschriften)

( Kontrollkästchen für Beschriftung aktivieren

Nachteil:  Konsolidierungsergebnisse werden bei Änderungen in den Quelldaten nicht automatisch aktualisiert.

b) dynamische Konsolidierung 
Aktiviert man das Kontrollkästchen „Verknüpfungen mit Quelldaten“, so werden die Konsolidierungsergebnisse bei Änderungen in den Quelldaten automatisch aktualisiert.

Das Ergebnis ist eine gegliederte Tabelle, deren zweite Ebene ausgeblendet ist.

Die erste Ebene enthält die konsolidierten Werte, die zweite Ebene die Quellwerte, die aus Verknüpfungsformeln gewonnen werden. Die konsolidierten Werte werden aus den zugehörigen Werten der zweiten Ebene berechnet z.B. in C32 steht  =SUMME(C29:C31) und in C29:  =Berlin!$B$9  in C30: =Hamburg!$B$7 und in C31:  =München!$B$8
Die dynamisch konsolidierte Tabelle besteht also aus lauter Formeln, wird daher auch jederzeit aktualisiert.

Nachteil: Will man für die neue Tabelle Gesamtsummen für die einzelnen Spalten berechnen, so muss man aufpassen, dass die Bezüge sich nur auf die einzelnen sichtbaren Zellen beziehen und nicht die ausgeblendeten Daten mit berücksichtigt werden, also =SUMME(C19;C22;C25;C28;C32) und nicht =SUMME(C19:C32).

( Übung 9
Hinweise zur Konsolidierung:

· Individuelle Anordnung der Zieltabelle 
Wollen Sie die Konsolidierung auf bestimmte Rubriken beschränken oder die Spalten und Zeilen anders anordnen, so erstellen Sie zunächst die Zeilen- und Spaltenbeschriftungen in der neuen Tabelle manuell. Dann markieren Sie den gesamten neuen Zielbereich inklusive der Be​schriftungen und führen dann die Konsolidierung mittels DATEN / KONSOLIDIEREN durch.

· Mehrere ähnliche Rubriken zu einer Rubrik zusammenfassen
Mit dem Stellvertreterzeichen * in einer Rubrik des Zielbereichs, kann man auch mehrere ähnliche Rubriken zusammenfassen. 
Beispiel: In den Quellbereichen kommen die Rubriken Autoreifen, Autositze und Autoradios vor, so können Sie alle drei mit der Rubrik Auto* im Zielbereich zusammenfassen.

· Konsolidierungen mittels Pivot-Tabelle 
Man kann Daten noch flexibler konsolidieren, indem man eine Pivot-Tabelle erstellt. Die Vorgehensweise ist dabei ähnlich zu dem hier vorgestellten Konsolidieren nach Rubrik.

Zelladressen oder Funktionsnamen werden nicht in Hochkommata eingeschlossen.


Soll davor ein Operatorzeichen wie > stehen, muss dieses in Hochkommata eingeschlossen werden und beides mit dem Zeichen & (Textverknüpfungsoperator) verbunden werden.  ">"&L9 





A1*A10�
B1*B10�
C1*C10�
�









1

